

Advanced Math Improvements in Maple 2023

Maple 2023 includes many improvements to the math engine. In addition to those described on this page, Maple 2023 includes a new package for [Quantifier Elimination](#) and updates to the [Graph Theory](#) package.

Better answers for indefinite integrals of algebraic functions

- The `int` command now uses Blake's heuristic to look for substitutions to find elementary forms for certain algebraic integrands. Blake's method works by considering algebraic function integrands of the form $\frac{p(x) \sqrt[m]{r(x)^n}}{q(x)}$ and then searching for substitutions (by solving systems of linear equations) that can put the integral into a simpler form, e.g. $\int \frac{p(x) \sqrt[m]{r(x)^n}}{q(x)} dx = \int \frac{v(u) \sqrt[m]{au+b}}{w(u)} du$, for which compact elementary answers can often be easily computed.
- Blake's heuristic has been available since Maple 2022 via the option `method=pseudoelliptic` but it is now tied directly into the default methods for computing integrals of algebraic functions.

Examples

- The following examples produced a very complicated expression in earlier versions of Maple, but now produce a simple, elementary answer.
- This is an example where in Maple 2022 the default answer comes from the elliptic integration method, and the answer is in terms of elliptic functions and is quite long:

```
> int( (x^3-2)*sqrt(-x^3-1)/(x^3+1)/(x^3+x^2+1), x, method=
      'elliptic' );
```

$$\frac{1}{3\sqrt{-x^3-1}} \left(2I\sqrt{3} \sqrt{I\left(x - \frac{1}{2} - \frac{I\sqrt{3}}{2}\right)\sqrt{3}} \sqrt{\frac{x+1}{\frac{3}{2} + \frac{I\sqrt{3}}{2}}} \right. \\ \left. \sqrt{-I\left(x - \frac{1}{2} + \frac{I\sqrt{3}}{2}\right)\sqrt{3}} \operatorname{EllipticF}\left(\frac{\sqrt{3} \sqrt{I\left(x - \frac{1}{2} - \frac{I\sqrt{3}}{2}\right)\sqrt{3}}}{3}, \right. \right. \\ \left. \left. \sqrt{\frac{I\sqrt{3}}{\frac{3}{2} + \frac{I\sqrt{3}}{2}}}\right) \right) + I\sqrt{2} \left(\right.$$

$$\sum_{\alpha = \text{RootOf}(_Z^3 + _Z^2 + 1)}$$

$$\frac{1}{\sqrt{-x^3 - 1}} \left(-\alpha \sqrt{\frac{1}{2}(-1 + 2x - I\sqrt{3})} \sqrt{\frac{x+1}{3 + I\sqrt{3}}} \sqrt{-\frac{1}{2}(-1 + 2x + I\sqrt{3})} \left(-\alpha^2 + 3 + I\sqrt{3}(-\alpha^2 - 2\alpha - 1) \right) \text{EllipticPi} \left(\frac{\sqrt{3} \sqrt{I \left(x - \frac{1}{2} - \frac{I\sqrt{3}}{2} \right) \sqrt{3}}}{3}, \frac{3\alpha^2}{2} \right. \right. \\ \left. \left. + 3\alpha - \frac{I\alpha^2\sqrt{3}}{2} + \frac{3}{2} + \frac{3I\sqrt{3}}{2}, \sqrt{\frac{I\sqrt{3}}{\frac{3}{2} + \frac{I\sqrt{3}}{2}}} \right) \right)$$

• In Maple 2023 the answer is elementary and compact.

> `int((x^3-2)*sqrt(-x^3-1)/(x^3+1)/(x^3+x^2+1), x);`

$$2 \operatorname{arctanh} \left(\frac{\sqrt{-x^3 - 1}}{x} \right)$$

• In Maple 2022 the following integral produced an answer in terms of elliptic functions and RootOfs of length 790119, but now the answer is elementary and very compact.

> `int((x^4 + x^3 + 1)*sqrt(-x^4 - 2*x^3 + 1)/((x^4 + 2*x^3 - 1)*(x^4 + 2*x^3 + x^2 - 1)), x);`

$$\operatorname{arctanh} \left(\frac{\sqrt{-x^4 - 2x^3 + 1}}{x} \right)$$

- Finally, here is a fairly long integral with a compact new answer in Maple 2023. `method=elliptic` shows the answer previously returned in Maple 2022.

```
> int( (x^3+2)*(112*x^6-4*x^5+3*x^4-224*x^3+4*x^2+112)*(2*x^3+x^2-2)^(1/2)/x^4/(16*x^6-8*x^5-x^4-32*x^3+8*x^2+16), x);
```

$$\frac{1}{8\sqrt{6-2\sqrt{2}}\sqrt{6+2\sqrt{2}}x^3} \left(95 \left(\left(\frac{4\sqrt{2x^3+x^2-2}(28x^3+53x^2-28)\sqrt{6-2\sqrt{2}}}{285} \right. \right. \right. \\ \left. \left. \left. + \operatorname{arctanh}\left(\frac{2\sqrt{2x^3+x^2-2}}{x\sqrt{6-2\sqrt{2}}}\right) x^3 \left(\sqrt{2} - \frac{124}{95}\right) \right) \sqrt{6+2\sqrt{2}} - \left(\sqrt{2} \right. \right. \\ \left. \left. + \frac{124}{95}\right) \operatorname{arctanh}\left(\frac{2\sqrt{2x^3+x^2-2}}{x\sqrt{6+2\sqrt{2}}}\right) x^3 \sqrt{6-2\sqrt{2}} \right) \right)$$

The response from Maple 2022 can be obtained using the following statement. Output is omitted from this document because it filled 25 pages.

```
> int( (x^3+2)*(112*x^6-4*x^5+3*x^4-224*x^3+4*x^2+112)*(2*x^3+x^2-2)^(1/2)/x^4/(16*x^6-8*x^5-x^4-32*x^3+8*x^2+16), x, method='elliptic');
```

Enhanced 'method' option for indefinite integrals

- The `int` command now exposes more internal routines for indefinite integration through the `method option`. It is now possible to directly call the integration by parts routine `Parts` and a parallel version of the Risch algorithm `ParallelRisch`.

```
> int(ln(exp(x)), x, method=parallelrisch);
```

$$-\frac{x^2}{2} + \ln(e^x)x$$

```
> int(4*x^2*arcsinh(x), x, method=parts);
```

$$\frac{4x^3 \operatorname{arcsinh}(x)}{3} - \frac{4x^2 \sqrt{x^2+1}}{9} + \frac{8\sqrt{x^2+1}}{9}$$

Better assumption support of MeijerG integrals

- The definite integration method that works via MeijerG convolutions now does a better job of checking conditions on parameters so that they are only applied under proper assumptions. As well, the methods will prompt via warning when the method could have produced an answer using additional assumptions.

```
> int( x*exp(-a*x^3 - b*x), x=0..infinity );
```

Warning, not enough assumptions for a closed form integral: a MeijerG transformation may be possible with additional assumptions: $a <> 0, 0 \leq \text{Re}(a), b <> 0, 0 \leq \text{Re}(b)$

$$\int_0^{\infty} x e^{-ax^3 - bx} dx$$

```
> assume(a::real, b::real);
```

```
> int( x*exp(-a*x^3 - b*x), x=0..infinity );
```

Warning, not enough assumptions for a closed form integral: a MeijerG transformation may be possible with additional assumptions: $0 < a, 0 < b$

$$\int_0^{\infty} x e^{-ax^3 - bx} dx$$

```
> assume(c>0, d>0);
```

```
> int( x*exp(-c*x^3 - d*x), x=0..infinity );
```

$$\frac{\sqrt{d} \left(\frac{d^{3/2}}{\sqrt{c}} \right)^{1/3} \left(-1 + \text{LommelS2} \left(1, \frac{2}{3}, \frac{2d^{3/2}\sqrt{3}}{9\sqrt{c}} \right) \right)}{3c^{5/6}}$$

```
> restart:
```

Support in solve for element-wise equations of rtables

- The `solve` command can now directly accept element-wise relations between vectors and matrices. Inputs of the form `M`, `M=c`, or `M=N` are transformed into a list of scalar equations (where `M` and `N` are rtables of the same dimensions and `c` is any scalar, and instead of `=` any binary relation of [type.relation](#) can be used).

```
> solve( (<<x+1|y+1>>,<z+2|w+2>>) = 1 );
```

$$\{w = -1, x = 0, y = 0, z = -1\}$$

- With no relations, a right-hand side of `0` is assumed

```
> solve( (<<x+1|y+1>>,<z+2|w+2>>) );
```

$$\{w = -2, x = -1, y = -1, z = -2\}$$

```
> solve( (<<x+1|y+1>>,<z+2|w+2>>) = (<<1|1>>,<2|2>>) );
```

$$\{w = 0, x = 0, y = 0, z = 0\}$$

```
> solve( (<<x+1|y+1>>,<z+2|w+2>>) < (<<1|1>>,<2|2>>) );
```

$$\{w < 0, x < 0, y < 0, z < 0\}$$

- While it is now possible to solve linear equations this way, it is advisable to use [LinearAlgebra:-LinearSolve](#) instead.

```

> n := 200:
> M := LinearAlgebra:-RandomMatrix(n):
> b := LinearAlgebra:-RandomVector(n):
> v := Vector( [ seq(cat(`v__`,i), i=1..n)]):
> CodeTools:-Usage( eval(v, solve( M.v = b )) ):
memory used=111.01MiB, alloc change=40.10MiB, cpu time=1.71s, real time=
1.62s, gc time=225.52ms

> CodeTools:-Usage( LinearAlgebra:-LinearSolve(M, b) ):
memory used=54.66MiB, alloc change=-1.02MiB, cpu time=1.01s, real time=
956.00ms, gc time=113.58ms

```

Polyhedral sets

- The [IntegerHull](#) command in the [PolyhedralSets](#) package has been extended to support higher dimensional polyhedral sets, both bounded and unbounded ones.
- The following example is an unbounded three-dimensional set.

```

> with(PolyhedralSets):
> ineqs := [-x[1] - 132/205*x[2] - 62/205*x[3] <= -1358/123,
           -x[1] + x[2] + x[3] <= 1405/17,
           x[1] - 6/59*x[2] + 83/177*x[3] <= 3500/59];
ineqs := ⎡ -x1 -  $\frac{132x_2}{205}$  -  $\frac{62x_3}{205}$  ≤ - $\frac{1358}{123}$ , -x1 + x2 + x3 ≤  $\frac{1405}{17}$ , x1 -  $\frac{6x_2}{59}$  +  $\frac{83x_3}{177}$ 
           ≤  $\frac{3500}{59}$  ⎤
> poly := PolyhedralSet(ineqs, [x[1], x[2], x[3]]);
poly := ⎧ Coordinates : [x1, x2, x3]
        Relations    : ⎡ -x1 -  $\frac{132x_2}{205}$  -  $\frac{62x_3}{205}$  ≤ - $\frac{1358}{123}$ , -x1 + x2 + x3 ≤  $\frac{1405}{17}$ , x1 -  $\frac{6x_2}{59}$  +  $\frac{83x_3}{177}$  ≤ -
> IsBounded(poly);
false

```

> IntegerHull(poly);

```
[[[-20, 36, 26], [-4, -25, 103], [-2, -30, 107], [-1, -36, 117], [0, -38, 118], [0, -36, 118], [1,
-37, 112], [1, -34, 117], [2, -39, 113], [2, -38, 114], [10, -43, 95], [26, -51, 60], [399, -238,
-776], [403, -240, -785], [453, -265, -897], [1544, -811, -3342]], [[101, 260, -159],
[12072, -6041, -27054], [-70, 267, -337]]]
```

• A bounded four-dimensional example.

```
> vertices := [[10, 10, 10, 10/3],
               [-140/8, -220/12, -10, -10/3],
               [60/8, 20, -100/12, -70/3],
               [-10/4, -100/12, 70/2, 35/3],
               [0, 0, 0, 50/3]];
```

```
vertices := [[10, 10, 10, 10/3], [-35/2, -55/3, -10, -10/3], [15/2, 20, -25/3, -70/3], [-5/2, -25/3,
35, 35/3], [0, 0, 0, 50/3]]
```

```
> vars := [x[1], x[2], x[3], x[4]]:
```

```
> poly := PolyhedralSet(vertices, [], vars);
```

```
poly := {
  Coordinates : [x1, x2, x3, x4]
  Relations   : [-x1 + 503x2/694 + 85x3/694 + 311x4/2082 ≤ 7775/3123, -x1 + 2715x2/2234 + 603x3/2234 + 813x4/2234 ≤
```

```
> IsBounded(poly);
```

true

```
> IntegerHull(poly);
```

```
[[[-15, -16, -6, -2], [-15, -15, -9, -4], [-14, -15, -4, -1], [-13, -13, -8, -1], [-13,
-12, -9, -5], [-12, -13, -4, 1], [-12, -13, -4, 2], [-12, -12, -3, -3], [-11, -12, -3,
-1], [-11, -11, -6, -3], [-11, -11, -1, -3], [-10, -8, -8, -5], [-9, -6, -8, -8],
[-7, -7, -4, 7], [-7, -6, -5, 3], [-7, -3, -8, -10], [-7, -3, -7, -10], [-6, -4, -5, 0],
[-5, -9, 23, 8], [-5, -4, -4, 5], [-5, -4, -3, -2], [-4, -5, 3, 10], [-4, -5, 3, 11], [-4,
-3, -2, -3], [-4, 0, -6, -9], [-3, -8, 30, 10], [-3, -7, 23, 10], [-3, -7, 23, 11], [-3, -6,
24, 6], [-3, 3, -8, -12], [-3, 3, -7, -13], [-2, -7, 31, 11], [-2, -6, 24, 8], [-2, -6, 24,
12], [-2, -6, 26, 11], [-2, -5, 25, 7], [-2, -5, 26, 6], [-2, -2, -1, 13], [-2, -1, -1, 1],
[-2, 3, -4, -8], [-1, -5, 25, 12], [-1, -5, 29, 9], [-1, -4, 19, 13], [-1, -4, 27, 7], [-1,
-3, 12, 13], [-1, -3, 12, 14], [-1, -1, 1, 14], [-1, -1, 2, 0], [-1, 1, 11, -4], [0, -2, 22, 7], [0,
-1, 8, 14], [0, -1, 20, 5], [0, 0, 1, 15], [0, 0, 2, 15], [0, 0, 13, 6], [0, 3, -3, 4], [0, 8, -8, -16], [1,
```

$-3, 27, 9], [1, -2, 23, 9], [1, 0, 8, 3], [1, 0, 22, 3], [1, 2, 0, 12], [1, 2, 5, 9], [1, 2, 9, 6], [1, 3, -1, 9],$
 $[1, 5, 5, -5], [1, 8, -3, -13], [2, 1, 9, 3], [2, 1, 21, 3], [2, 2, 2, 14], [2, 2, 4, 1], [2, 5, 7, 0], [2, 5, 13,$
 $-5], [2, 10, -7, -13], [2, 11, -8, -17], [3, 7, -2, 3], [3, 7, 10, -6], [3, 7, 11, -7], [3, 8, -3, 0],$
 $[3, 8, 6, -7], [3, 11, -2, -15], [4, 2, 17, 6], [4, 8, 8, -5], [4, 9, -2, -3], [4, 10, -1, -5], [4, 10,$
 $3, -8], [4, 11, -3, -13], [4, 11, -2, -13], [4, 12, -5, -15], [4, 12, 0, -15], [4, 14, -7, -19],$
 $[5, 3, 18, 6], [5, 3, 18, 7], [5, 3, 19, 6], [5, 5, 6, 3], [5, 13, -5, -10], [5, 13, 1, -14], [5, 14, -6,$
 $-13], [5, 15, -7, -18], [5, 15, -7, -17], [6, 9, 3, -4], [6, 15, -4, -14], [6, 16, -6, -16], [6,$
 $17, -7, -20], [7, 6, 14, 5], [7, 12, 1, -7], [7, 13, 0, -10], [7, 13, 2, -9], [7, 13, 3, -10], [7, 15,$
 $-3, -12], [7, 16, -4, -16], [7, 17, -4, -18], [7, 18, -6, -20], [8, 8, 8, 6], [8, 8, 9, 3], [8, 8, 12,$
 $3], [8, 12, 3, -4], [9, 10, 8, 1], [9, 10, 8, 2], [9, 10, 9, 1], [9, 13, 4, -5]], []]$

New support for Z-polyhedral sets

- The [ZPolyhedralSets](#) subpackage is new in Maple 2023. It is part of the [PolyhedralSets](#) package and deals with Z-polyhedral sets. A Z-polyhedral set is the intersection of a polyhedral set with an integer lattice.


```
> with(PolyhedralSets): with(ZPolyhedralSets):
```

- Here is a three-dimensional polyhedral set.

```
> ineqs := [0 <= -16 + 2*y + z, 0 <= -72 + 4*x + 4*y + 3*z, 0 <= 2*y -
z, 0 <= -24 + 4*x + 4*y - 3*z, 0 <= -4*x + 4*y + 3*z, 0 <= 48 - 4*x
+ 4*y - 3*z, 0 <= 48 - 4*x - 4*y + 3*z, 0 <= 8 - 2*y + z, 0 <= -24 +
4*x - 4*y + 3*z, 0 <= 24 - 2*y - z, 0 <= 24 + 4*x - 4*y - 3*z, 0 <=
96 - 4*x - 4*y - 3*z]:
```

```
> ps := PolyhedralSet(ineqs):
```

```
> Plot(ps);
```




- Now we define the Z-polyhedral set of all integer lattice points in ps .

```
> psz := ZPolyhedralSet(ineqs, [x, y, z]):
```



```
> PlotIntegerPoints3d(psz);
```



Improvements to parametric limits on branch cuts

- The option `parametric` to the `limit` command has been extended to handle more cases where the expansion point may lie on the branch cut of a mathematical function for some real parameter values. Examples:

```
> unassign('a');
```

```
> limit(ln(a+I*x), x=0, 'right', 'parametric');
```

$$\left\{ \begin{array}{ll} \ln(-a) + I\pi & a < 0 \\ -\infty & a = 0 \\ \ln(a) & 0 < a \\ \lim_{x \rightarrow 0^+} \ln(a + Ix) & \text{otherwise} \end{array} \right.$$

```
> limit(arccoth(a+I*x), x=0, 'right', 'parametric');
```

$$\lim_{x \rightarrow 0^+} \operatorname{arccoth}(a + Ix) \begin{cases} -\operatorname{arccoth}(-a) & a < -1 \\ -\infty & a = -1 \\ \operatorname{arctanh}(a) - \frac{I\pi}{2} & a < 1 \\ \infty & a = 1 \\ \operatorname{arccoth}(a) & 1 < a \\ \operatorname{arccoth}(a + Ix) & \text{otherwise} \end{cases}$$

```
> limit(Li(a+I*x), x=0, 'right', 'parametric');
```

$$\lim_{x \rightarrow 0^+} \operatorname{Li}(a + Ix) \begin{cases} \operatorname{Ei}(\ln(-a) + I\pi) & a < 0 \\ I\pi & a = 0 \\ I\pi + \operatorname{Ei}(\ln(a)) & a < 1 \\ -\infty & a = 1 \\ \operatorname{Ei}(\ln(a)) & 1 < a \\ \operatorname{Li}(a + Ix) & \text{otherwise} \end{cases}$$

New certified univariate complex solver

- A new, faster method for isolating the complex roots of a univariate polynomial with *complex(numeric)* coefficients has been added to the [RootFinding:-Isolate](#) command. It is accessible through the option **complex** or **method=HR** and supports most of the existing options of [RootFinding:-Isolate](#), as well as a new option **domain** for limiting the search space. See the corresponding help page for more details. Examples:

```
> with(RootFinding):
```

```
> Isolate(x^6-1, x, 'complex');
```

```
[x = -1., x = 1., x = 0.5000000000 + 0.8660254038 I, x = 0.5000000000 - 0.8660254038 I, x = -0.5000000000 + 0.8660254038 I, x = -0.5000000000 - 0.8660254038 I]
```

```
> Isolate(x^2-I, x, 'complex', 'output=interval');
```

$$x = \left[\frac{854839645001008256604305}{1208925819614629174706176} + \frac{854839645001008256604305 I}{1208925819614629174706176}, \frac{854839645001010290403183}{1208925819614629174706176} + \frac{854839645001010290403183 I}{1208925819614629174706176} \right], x = \left[-\frac{854839645001010290403183}{1208925819614629174706176} - \frac{854839645001010290403183 I}{1208925819614629174706176}, -\frac{854839645001008256604305}{1208925819614629174706176} - \frac{854839645001008256604305 I}{1208925819614629174706176} \right]$$

```
> Isolate(x^2-I, x, 'complex', 'digits'=100);
```

$$x = 0.7071067811865475244008443621048490392848359376884740365883398689953662392310535194251937671638207864 + 0.7071067811865475244008443621048490392848359376884740365883398689953662392310535194251937671638207864 I, x = -0.7071067811865475244008443621048490392848359376884740365883398689953662392310535194251937671638207864 - 0.7071067811865475244008443621048490392848359376884740365883398689953662392310535194251937671638207864 I]$$

- Find 10 roots in the first quadrant.

```
> Isolate(x^100+x^3-1, x, 'method=HR', 'domain'=[0,infinity+I*infinity], 'maxroots'=10);
```

$$x = 0.9744076649, x = 0.9781052241 + 0.1737357669 I, x = 0.9531760998 + 0.2980270829 I, x = 0.9680195472 + 0.2362266682 I, x = 0.9104543811 + 0.4181438477 I, x = 0.9827791401 + 0.1111417351 I, x = 0.9804852800 + 0.05036970385 I, x = 0.1198693017 + 0.9978851093 I, x = 0.4848690063 + 0.8826473106 I, x = 0.2461381235 + 0.9755862161 I]$$

New method option for fsolve, for a univariate polynomial

- The `fsolve` command has been updated with a new `method` option. When solving for roots of a univariate polynomial this option allows specification of the new solver within the `RootFinding:-Isolate` command, or the original modified Laguerre method of the NAG Library. The value for the option can be specified here as `method=hefroots` or `method=NAG`. The current default is `method=NAG`, unless `fsolve` determines that the problem is ill-conditioned.

```
> [ fsolve(x^4-5.591265174*x^3+2.936595076*x^2+17.11912543*
      x-10.52786025, x=0..4, method=hefroots) ];
      [0.6180339885, 2.718281830, 3.872983344]
```

New method option for fsolve, for a univariate non-polynomial equation

- When solving for roots of an equation in one variable that is not a polynomial, the new `method` option lets you specify the method used. The value for the option can be specified here as `method=subdivide` or `method=NextZero`. This option must be accompanied by the `maxsols` option. The `subdivide` method repeatedly searches on new subintervals between roots already found, using a custom mix of approaches on each. The `NextZero` method utilizes the [RootFinding:-NextZero](#) command to repeatedly search past the last root found. The default is `method=subdivide`, which may be more suitable for equations that are not continuous or have non-isolated roots. For some problems the `NextZero` method may be significantly faster.

```
> [ fsolve(BesselJ(0,x^3)-1/100*x^2-floor(x)+2, x=0..4, maxsols=10,
      method=subdivide) ];
      [2.040911370, 2.290794873, 2.446318006, 2.642712321, 2.748619065, 2.920713827, 2.995225008]
```

```
> [ fsolve(BesselJ(0,x^3)-1/100*x^2, x=0..4, maxsols=10, method=
      NextZero) ];
      [1.333416382, 1.777268850, 2.040911369, 2.290794872, 2.446318005, 2.642712321, 2.748619064,
      2.920713827, 2.995225007, 3.155131991]
```

New support for lazy multivariate Puiseux series

- The [MultivariatePowerSeries](#) package was introduced in Maple 2021. It deals with multivariate power series in a lazy fashion, meaning that extra terms for a given result can be computed quickly.
- In Maple 2023, this package can now deal with multivariate Puiseux series and univariate polynomials over multivariate Puiseux series.

```
> with(MultivariatePowerSeries):
```

> a := PowerSeries(exp(s + t)*ln(1+t));

$$a := \left[\text{PowerSeries of } e^s e^t \ln(1+t) : t + \frac{t^2}{2} + st + \frac{t^3}{3} + \frac{st^2}{2} + \frac{s^2 t}{2} + \frac{st^3}{3} + \frac{s^2 t^2}{4} + \frac{s^3 t}{6} + \frac{3t^5}{40} + \frac{s^2 t^3}{6} + \frac{s^3 t^2}{12} + \frac{s^4 t}{24} + \dots \right]$$

> b := PuiseuxSeries(a, [s = sqrt(x), t = y^(1/3)*x]);

$$b := \left[\text{PuiseuxSeries of } e^{\sqrt{x}} e^{y^{1/3}x} \ln(1 + y^{1/3}x) : y^{1/3}x + \frac{y^{2/3}x^2}{2} + x^{3/2}y^{1/3} + \frac{yx^3}{3} + \frac{x^{5/2}y^{2/3}}{2} + \frac{x^2y^{1/3}}{2} + \frac{x^{7/2}y}{3} + \frac{x^3y^{2/3}}{4} + \frac{x^{5/2}y^{1/3}}{6} + \frac{3y^{5/3}x^5}{40} + \frac{x^4y}{6} + \frac{x^{7/2}y^{2/3}}{12} + \frac{x^3y^{1/3}}{24} + \dots \right]$$

> Truncate(1/b, 3, mode=absolute);

$$-\frac{1}{2} + \frac{1}{xy^{1/3}} - \frac{1}{\sqrt{x}y^{1/3}} + \frac{1}{2y^{1/3}} - \frac{\sqrt{x}}{6y^{1/3}} + \frac{\sqrt{x}}{2} + \frac{x}{24y^{1/3}} - \frac{x}{4} - \frac{x^{3/2}}{120y^{1/3}} - \frac{y^{1/3}x}{12} + \frac{x^{3/2}}{12} + \frac{x^2}{720y^{1/3}} + \frac{x^{3/2}y^{1/3}}{12} - \frac{x^2}{48} - \frac{x^{5/2}}{5040y^{1/3}} - \frac{x^2y^{1/3}}{24} + \frac{x^{5/2}}{240} + \frac{5y^{2/3}x^2}{24} + \frac{x^{5/2}y^{1/3}}{72} - \frac{x^3}{1440}$$

- The package can also compute a Puiseux factorization of a univariate polynomial over univariate Puiseux series. We set this up by defining a univariate Puiseux series.

> c := PowerSeries(exp(t) * ln(1+t));

$$c := \left[\text{PowerSeries of } e^t \ln(1+t) : t + \frac{t^2}{2} + \frac{t^3}{3} + \frac{3t^5}{40} + \dots \right]$$

> d := PuiseuxSeries(c, [t = y^(1/3)]);

$$d := \left[\text{PuiseuxSeries of } e^{y^{1/3}} \ln(1 + y^{1/3}) : y^{1/3} + \frac{y^{2/3}}{2} + \frac{y}{3} + \frac{3y^{5/3}}{40} + \dots \right]$$

- We now define e as $dz^2 + \left(\frac{1}{d}\right)z + 1$ and compute its Puiseux factorization.

> e := UnivariatePolynomialOverPuiseuxSeries([d, 1/d, PuiseuxSeries(1), z]);

$$e := \left[\text{UnivariatePolynomialOverPuiseuxSeries: } \left(y^{1/3} + \frac{y^{2/3}}{2} + \frac{y}{3} + \frac{3y^{5/3}}{40} + \dots \right) + \left(\frac{1}{y^{1/3}} + \dots \right) z + (1) z^2 \right]$$

> e_factors := PuiseuxFactorize(e, returnleadingcoefficient=false);

```
> for f in e_factors do print(Truncate(f, 4)); end do:
```

$$1 - \frac{y^{1/3}}{2} - \frac{y^{2/3}}{12} - \frac{19y}{24} - \frac{829y^{4/3}}{720} + y^{1/3}z$$

```
> Truncate(e - mul(f, f in e_factors), 6);
```

0

- We see that the factors are linear or constant in z and they multiply to e .